/+ dub.sdl:
    name "A"
    dependency "dcomp" version=">=0.4.0"
+/

import std.stdio, std.algorithm, std.conv, std.range;
import std.typecons, std.math;

// import dcomp.scanner, dcomp.functional;

alias E = Tuple!(int, "to");

int U;
long[2][] d;
long[][][] dp;
bool[][][] used;
long calc(int p, int a, int c) {
    if (used[c][p][a]) return dp[c][p][a];
    used[c][p][a] = true;
    if (p == d.length) return 0;
    long[2] nw = d[p];
    int b = p-a;

    long ans = 10L^^18;
    if (c == 0) {
        ans = min(ans, U*(nw[0]+nw[1]) + calc(p+1, a, 1));
    }
    if (a < U) {
        ans = min(ans, a*(nw[0]+nw[1]) + nw[1] + calc(p+1, a+1, c));
    }
    if (b < U) {
        ans = min(ans, b*(nw[0]+nw[1]) + nw[0] + calc(p+1, a, c));
    }
    return dp[c][p][a] = ans;
}

int main() {
    dp = new long[][][](2, 5005, 2505);
    used = new bool[][][](2, 5005, 2505);
    auto sc = new Scanner(stdin);
    int n;
    sc.read(n);
    d = new long[2][n];
    foreach (i; 0..n) {
        sc.read(d[i]);
    }
    d.sort!"a[0]+a[1] > b[0]+b[1]";
    U = n/2;
    writeln(calc(0, 0, 1-n%2));
    return 0;
}
/* IMPORT /home/yosupo/Program/dcomp/source/dcomp/functional.d */
// module dcomp.functional;

struct memoCont(alias pred) {
    import std.range, std.algorithm, std.conv;
    import std.string : join;
    import std.traits : ReturnType, ParameterTypeTuple, isIntegral;
    import std.typecons : tuple, Tuple;
    import std.meta;
    alias R = ReturnType!pred;
    alias Args = ParameterTypeTuple!pred;
    static assert (allSatisfy!(isIntegral, Args)); // should be int only?

    static immutable N = Args.length;
    static string toTuple(string s) { //int[N] -> (N[0], N[1], N[2], ...)
        return "(" ~ iota(N).map!(i => s~"["~i.to!string~"]").join(",") ~ ")";
    }
    static string toArray(string s) { //int[N] -> [N[0]][N[1]][N[2]]...
        return iota(N).map!(i => "["~s~"["~i.to!string~"]]").join("");
    }
    template NArray(T, int N) {
        static if (!N) alias NArray = T;
        else alias NArray = NArray!(T, N-1)[];
    }
    int[2][N] rng;
    NArray!(R, N) dp;
    NArray!(bool, N) used;
    void init(int[2][N] rng) {
        this.rng = rng;
        int[N] len = rng[].map!(a => a[1]-a[0]+1).array;
        
        //dp = new typeof(dp)(len[0], len[1], ..., len[N-1])
        //used = new typeof(used)(len[0], len[1], ..., len[N-1])
        dp = mixin("new typeof(dp)"~toTuple("len"));
        used = mixin("new typeof(used)"~toTuple("len"));
    }
    R opCall(Args args) {
        int[N] idx;
        foreach (i, v; args) {
            assert(rng[i][0] <= v && v <= rng[i][1]);
            idx[i] = v - rng[i][0];
        }
        //if (used[idx[0]]..[idx[N-1]]) dp[idx[0]]..[idx[N-1]]
        //used[idx[0]]..[idx[N-1]] = true
        if (mixin("used"~toArray("idx"))) return mixin("dp"~toArray("idx"));
        mixin("used"~toArray("idx")) = true;

        auto r = pred(args);

        //dp[idx[0]]..[idx[N-1]] = r
        mixin("dp"~toArray("idx")) = r;

        return r;
    }
}
/* IMPORT /home/yosupo/Program/dcomp/source/dcomp/scanner.d */
// module dcomp.scanner;

class Scanner {
    import std.stdio : File;
    import std.conv : to;
    import std.range : front, popFront, array, ElementType;
    import std.array : split;
    import std.traits : isSomeChar, isStaticArray, isArray; 
    import std.algorithm : map;
    File f;
    this(File f) {
        this.f = f;
    }
    string[] buf;
    private bool succ() {
        while (!buf.length) {
            if (f.eof) return false;
            buf = f.readln.split;
        }
        return true;
    }
    private bool readSingle(T)(ref T x) {
        if (!succ()) return false;
        static if (isArray!T) {
            alias E = ElementType!T;
            static if (isSomeChar!E) {
                //string or char[10] etc
                x = buf.front;
                buf.popFront;
            } else {
                static if (isStaticArray!T) {
                    //static
                    assert(buf.length == T.length);
                }
                x = buf.map!(to!E).array;
                buf.length = 0;                
            }
        } else {
            x = buf.front.to!T;
            buf.popFront;            
        }
        return true;
    }
    int read(T, Args...)(ref T x, auto ref Args args) {
        if (!readSingle(x)) return 0;
        static if (args.length == 0) {
            return 1;
        } else {
            return 1 + read(args);
        }
    }
}

unittest {
    import std.path : buildPath;
    import std.file : tempDir;
    import std.algorithm : equal;
    import std.stdio : File;
    string fileName = buildPath(tempDir, "kyuridenanmaida.txt");
    auto fout = File(fileName, "w");
    fout.writeln("1 2 3");
    fout.writeln("ab cde");
    fout.writeln("1.0 1.0 2.0");
    fout.close;
    Scanner sc = new Scanner(File(fileName, "r"));
    int a;
    int[2] b;
    char[2] c;
    string d;
    double e;
    double[] f;
    sc.read(a, b, c, d, e, f);
    assert(a == 1);
    assert(equal(b[], [2, 3]));
    assert(equal(c[], "ab"));
    assert(equal(d, "cde"));
    assert(e == 1.0);
    assert(equal(f, [1.0, 2.0]));
}


Measured memory_bytes: 242556